

# A Trading System development tutorial for the AmiBroker Stock Charting/Analysis program

By [Herman van den Bergen](#), Monday, July 29, 2002.

This tutorial was prepared in response to a post by Dimitris Tsokakis, on the AmiBroker Users list, in which he introduces the basic idea for a StoCCI system (see text on page 2).

*The procedures used below reflect personal design philosophy and preferences. I do not use any commissions or interest in my calculations, if these play a role the system it is so weak that I discard it. You may or may not agree with my approach, either way I encourage you to express your opinion on the [AmiBroker list](#). It is only through frank dialogue and criticism that we can improve our skills. Kindly report any errors or omissions to me privately: [psytek@rogers.com](mailto:psytek@rogers.com)*

**“The Starting Position”**  
**Profits: 143% for the period 4/12/00 to 7/26/02**



**Important Note:** This report outlines typical procedures that you may use to develop a trading system. **Do not trade this system as is; no profitability is suggested or implied.**

## Dimitris' Post

Posted on [amibroker@yahoo.com](mailto:amibroker@yahoo.com)

By: Dimitris Tsokakis [[TSOKAKIS@ONEWAY.GR](mailto:TSOKAKIS@ONEWAY.GR)]

Subject: [amibroker] Tranformations

Date: Mon 7/29/02 5:41 AM

*Suppose you trade INTC with a Stochastic CCI [6,12,70] system.  
The total % net profit, after 18trades/12 winners/6losers is -15.85%  
See INTC1.gif  
[buy,sell,short,cover at open, delay +1, commission 0.5%, all stops disabled] with the use of the code:*

```
/*Smoothed Stochastic CCI*/  
P =6;D=DateNum(>1000301);  
Blevel =Optimize("BL",12,10,20,1);Slevel = Optimize("SL",70,70,90,1);  
StOcci=100*(CCI(6)-LLV(CCI(6),14))/(HHV(CCI(6),14)-LLV(CCI(6),14));  
stocci=MA(stocci,5);  
Buy=Cross(STOCCI,BLEVEL);Sell= Cross(STOCCI, Slevel);  
Buy=D*(Buy);Sell=D*(Sell);Short = Sell;Cover = Buy;  
Buy=ExRem(Buy,Sell);Sell=ExRem(Sell,Buy);  
Short=ExRem(Short,Cover);Cover=ExRem(Cover,Short);  
E1=Equity();EM40=MA(E1,40);  
XB=Flip(Buy,Sell);XS=Flip(Sell,Buy);  
XSH=2*Flip(Short,Cover);XCO=2*Flip(Cover,Short);  
XBSH=(XB+XSH);  
XSXCO=(XS+XCO);  
ASC=Cross(E1,EM40);DESC=Cross(EM40,E1);  
// Buy=ASC*(XBSH==1);Sell=DESC*(XSXCO==1);  
//Short=ASC*(XBSH==2);Cover=DESC*(XSXCO==2);
```

*Do you want to transform the -15.85% to a +126% somehow automatically ?*

*Simply delete the // from the last line !!*

*6trades/5winners/1loser and that's it.*

*See INTC2.gif*

*Dimitris Tsokakis*

*PS1. Any volunteers to improve this method and reach its final form with a co-operative work ??*

*We may create a method belonging to this excellent list.*

*This is my ambition. I know what to do for myself. With or, sometimes, without strict coding.*

*PS2. The above combination [6,12,70] is just for the example*

*PS3. Some other examples*

*CSCO from +39% to +364%*

*MSFT from -11% to +87%*

*QQQ from +158% to +328%*

*but*

*ORCL from +36% to +9% etc. etc*

Thank you Dimitris, for this and many other interesting posts!

## ***The Tools***

- 1) [AmiBroker](#) EOD 4.07 beta (version 4.07.4, July 28,2002)
- 2) Excel 2000 (3D charts, I don't know if 3D macro works in other versions)
- 3) Excel Plugin: [ABOptimizer392c.zip](#) (Makes 3D charts for you)
- 4) [AmiBroker User group](#) (here you can join discussions and find support)

## ***Preliminary observations***

- There are three periods that can be optimized
- The DateNum() has to be changed to allow out-of-sample optimizing
- The usual smoothing of the Stochastic formula has been disabled
- The LookBack period is rather large
- Crossovers are not as usual and may need optimizing
- Before adding signal gating (Equity Feedback) the basic formula should be optimized
- No Stops are used
- ExRem statements may not be needed
- Equity(1) may be needed to ensure proper Buy/Sell signals
- The gating MA() uses a very long period of 40



## ***Selecting a test period***

There are many ways to do this however it is my personal preference to develop on data (starts 1/1/1997 for me) before 1/1/01 and perform out-of-sample testing on the subsequent period of 1/1/01 to today. I will adopt these choices for this project.

I disable the starting date by changing the second line of code  
(**D=DateNum()**>1000301;) to (**D= 1; //DateNum()**>1000301;).

## ***The System test Settings***

To duplicate this exercise you will need INTC data from 1/1/97 to today. I use QuotePlus data, if you use another data source your results may be slightly different. Do not be concerned about that; it is the procedures that are important, not the exact figures or profits. Before starting you must make INTC your current stock, select it in your workspace. I use these three testing periods:



## Crossover Optimizing

When working with new systems I always first determine the optimum direction of signal-threshold crossings. This is a reversal system so there are only two ways the StoCCI indicator can cross the thresholds.

If the system were utilizing cash positions there would be four possible combinations. I perform this step on “All Quotations” because the system, at this stage, performs very poorly; using a longer testing period will give me a better indication. I assume that the optimum crossings remain constant for the system.

For this test I also set the Threshold levels to “neutral” that would be 50 in this case. I do this to prevent distortion of the results by far-out threshold choices. I use this code to test the crossovers:

```
/*Listing #1 - Smoothed Stochastic CCI*/
TM = Optimize("TML",1,1,2,1); //Determine optimum direction of crossing
P = 6; //Optimize("Pd",6,2,10,1); //Periods as provided by Dimitris
LB = 14; //Optimize("LB",14,1,20,1);
Ps = 1; //Optimize("Ps",5,1,6,1);
D = 1; //DateNum(>1000301); //Disable datenum()
Blevel = 50; //Optimize("BL",12,10,20,1); //Use neutral thresholds
Slevel = 50; //Optimize("SL",70,70,90,1);

StOcci=100*(CCI(P)-LLV(CCI(P),LB))/(HHV(CCI(P),LB)-LLV(CCI(P),LB));
Plot(StoCCI,"StoCCI",4,1);
StoCCI=MA(stocci,Ps);

Buy = Iif(TM==1,Cross(STOCCI,BLEVEL),Cross(BLEVEL,STOCCI));
Sell = Iif(TM==1,Cross(Slevel,STOCCI),Cross(STOCCI,Slevel));
Buy = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;
```

The clear winner, which I confirmed with tests over other periods (not shown) is TM=2. Modifying the basic formula we now have:

```
/* Listing #2 - Smoothed Stochastic CCI*/
P = 6; //Optimize("Pd",6,2,10,1); //Periods as provided by Dimitris
LB = 14; //Optimize("LB",14,1,20,1);
Ps = 1; //Optimize("Ps",5,1,6,1);
D = 1; //DateNum(>1000301); //Disable datenum()
Blevel = 50; //Optimize("BL",12,10,20,1); //Use neutral thresholds
Slevel = 50; //Optimize("SL",70,70,90,1);

StOcci=100*(CCI(P)-LLV(CCI(P),LB))/(HHV(CCI(P),LB)-LLV(CCI(P),LB));
Plot(StoCCI,"StoCCI",4,1);
StoCCI=MA(stocci,Ps);

Buy = Cross(BLEVEL,STOCCI);
```

```

Sell = Cross(STOCCI,SLevel);
Buy  = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;

```

## Period Optimizing

All optimizations are performed on data from 1/1/1997 to 1/1/01. This reserves data from 1/1/01 to today for out-of-sample testing.

We have three (I am including Stochastic Smoothing which was disabled by Dimitris) that can be optimized. We can chart only two parameters vs. the gain at a time on a 3D chart so I will need to make many charts to find the optimum parameters. I decided to make a different chart for each Smoothing period (Ps). The two parameters to be charted in each case are the CCI Period (P) and the Look Back period (LB). I use the formula from Listing #2. Thus, Ps is changed manually while P and LB are changed by the AA Optimizer. The charts produced are shown on the next few pages. There is one chart for Ps=1, one for Ps=2, etc.

Looking at the charts and pick the best periods is not easy. Clearly the highest gain is obtained with Ps of 10, 11 and 12. We also note that the CCI Period (P) is 5 for each of these three cases. So, to start I'll pick the middle value of 11 for smoothing and use a period of 5 for the of CCI().

We still have three choices (8, 3 and 6) for the LookBack value. Further testing is needed to make a decision on this. To get a good feel of how they work I overlay the equities in one display, I use the following code to do this.

Text continues below the next few pages of 3D charts.




Figure 1 - Ps=1




Figure 2 - Ps=2




Figure 3 - Ps=3




Figure 4 - Ps=4




Figure 5 - Ps=5




Figure 6 - Ps=6




Figure 7 - Ps=7






Figure 8 - Ps=8




Figure 9 - Ps=9




Figure 10 - Ps=10




Figure 11 - Ps=11




Figure 12 - Ps=12




Figure 13 - Ps=13




Figure 14 - Ps=14




Figure 15 - Ps=15

*/\* Listing #3 - Triple equities  
 /\* Smoothed Stochastic CCI\*/*

*// System #1*

*P = 5; //Optimize("Pd",6,2,10,1); //Periods as provided by Dimitris*

*LB = 8; //Optimize("LB",14,1,20,1);*

*Ps = 11; //Optimize("Ps",5,1,6,1);*

*D = 1; //DateNum()>1000301; //Disable datenum()*

*Blevel = 50; //Optimize("BL",12,10,20,1); //Use neutral thresholds*

*Slevel = 50; //Optimize("SL",70,70,90,1);*

*StOcci=100\*(CCI(P)-LLV(CCI(P),LB))/(HHV(CCI(P),LB)-LLV(CCI(P),LB));*

*StoCCI=MA(stocci,Ps);*

*Buy = Cross(BLEVEL,STOCCD);*

*Sell = Cross(STOCCI,SLevel);*

*Buy = D\*Buy;*

*Sell = D\*Sell;*

*Short = Sell;*

Author: [Herman van den Bergen](#)

```

Cover = Buy;
E1=Equity();
Plot(E1,"E1-LB=6 (RED)",4,1);

// System #2
/* Listing #2 - Smoothed Stochastic CCI*/
P = 5; //Optimize("Pd",6,2,10,1); //Periods as provided by Dimitris
LB = 3; //Optimize("LB",14,1,20,1);
Ps = 11; //Optimize("Ps",5,1,6,1);
D = 1; //DateNum()>1000301; //Disable datenum()
Blevel = 50; //Optimize("BL",12,10,20,1); //Use neutral thresholds
Slevel = 50; //Optimize("SL",70,70,90,1);

StOcci=100*(CCI(P)-LLV(CCI(P),LB))/(HHV(CCI(P),LB)-LLV(CCI(P),LB));
StoCCI=MA(stocci,Ps);

Buy = Cross(BLEVEL,STOCCI);
Sell = Cross(STOCCI,SLevel);
Buy = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;
E2=Equity();
Plot(E2,"E2 LB=8 (BLK)",1,1);

//System #3
/* Listing #2 - Smoothed Stochastic CCI*/
P = 5; //Optimize("Pd",6,2,10,1); //Periods as provided by Dimitris
LB = 6; //Optimize("LB",14,1,20,1);
Ps = 11; //Optimize("Ps",5,1,6,1);
D = 1; //DateNum()>1000301; //Disable datenum()
Blevel = 50; //Optimize("BL",12,10,20,1); //Use neutral thresholds
Slevel = 50; //Optimize("SL",70,70,90,1);

StOcci=100*(CCI(P)-LLV(CCI(P),LB))/(HHV(CCI(P),LB)-LLV(CCI(P),LB));
StoCCI=MA(stocci,Ps);

Buy = Cross(BLEVEL,STOCCI);
Sell = Cross(STOCCI,SLevel);
Buy = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;
E3=Equity();
Plot(E3,"E3 LB=6 (BLU)",6,1);

GraphXSpace = 5;
/*

```

## Equity Charts for LB=8, 3 and 6

The following chart show three equities for the 8, 3 and 6 LookBack periods. These systems were optimized *before* the vertical purple line; to the right of the Purple line is the out-of-sample period.



Figure 16 - Equities for basic system using LB=8(RED), 3(BLK) and 6(BLU).

To me this is a no-brainer, I'll continue with a look back period of 8. It is very encouraging to note the good response in the out-of-sample period after 1/1/01. The system need more work but looks promising.

### ***Threshold Optimization***

We now are working with a modified formula that gives about 5500% profits since 1/1/97 and 280% for the out of sample period 1/1/01 to 7/26/02. It is time to optimize the thresholds and I will use this formula:

```

/* Listing #4
/* Smoothed Stochastic CCI*/
D = 1; //DateNum()>1000301; //Disable datenum()
Blevel = Optimize("BL",49,10,20,1); //Use neutral thresholds
Slevel = Optimize("SL",49,70,90,1);
StOcci=100*(CCI(5)-LLV(CCI(5),3))/(HHV(CCI(5),3)-LLV(CCI(5),3));

```

```

StoCCI=MA(stocci,11);

Buy = Cross(BLEVEL,STOCCI);
Sell = Cross(STOCCI,SLevel);
Buy = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;
E1=Equity();
Plot(E1,"Equity Listing #4",4,1);
GraphXSpace = 5;
/*

```

I first optimize the thresholds separately, to save time. Doing this I obtain 49 for each, the Long and the Short Thresholds. Next I optimize them simultaneously to make sure there is no interaction between the thresholds – while I can't explain it I have seen this happen. Below are the Flat and 3D charts for thresholds from 25-75.

The 3D chart reveals a central point of high profits. Normally such sharp peaks worry me however in this case there are very few competitive peaks so it appears that this stock has some characteristics that make it response favorable to our system.


Also note that this chart uses a linear axis for the gain while the profits were compounded. I tried a log scale but it wasn't as clear to see the system's performance, so I show you the linear chart.

All in all I am pleased with the gain concentration around our thresholds of 49/49, so I'll continue working with these.


Our system now gives about 2100% profit in the development period (1997-2001) and gives about 400% profit during the out-of-sample period. Total profits since 1/1/1997 are about 12400%.

Below are the 4D charts for the Threshold optimizations.

### StoCCI Threshold Optimization



### StoCCI Threshold Optimization



## Adding Profit Stops

I believe that Stops are more time sensitive than periods and should be optimized using more recent data. This is my personal preference. Contrary to what many traders say I believe that Stops must be optimized separately for Long and Short positions. I modified the formula to do just that.

Again, to save time, I perform a quick optimization at 1% increments I find that 8% for Long and 9% for Short positions increase the returns. I have found that fractional percentages can make a lot of difference so I will repeat the optimization with 0.1% increments over a narrower range. My formula:

```
/* Listing #5
/* Smoothed Stochastic CCI*/
D = 1; //DateNum(>1000301; //Disable datenum()
StoCCI=100*(CCI(5)-LLV(CCI(5),3))/(HHV(CCI(5),3)-LLV(CCI(5),3));
StoCCI=MA(stoCCI,11);
Buy = Cross(49,STOCCI);
Sell = Cross(STOCCI,49);
Buy = D*Buy;
Sell = D*Sell;
Short = Sell;
Cover = Buy;

E = Equity();
Plot(E,"Equity wo Stops",4,1);

LPos = Flip(Buy,Sell);
SPos = Flip(Short,Cover);
LPT = 8.1; //Optimize("LProfitTarget",8,5,10,0.1);
SPT = 9.6; //Optimize("SProfitTarget",9,5,10,0.1);
PT = IIf(LPos,LPT,IIf(SPos,SPT,0));
ApplyStop(1,1,PT,1);

E = Equity();
Plot(E,"Equity with PT Stops",6,1);
GraphXSpace = 5;
/*
```

Profits increased from about 12K% to 20K% for the 1997-2002 range. Most of the profit increase took place in the out-of-sample period, see chart below.







Figure 17 - Blue = Equity with PT Stops, Red = No Stops

This looks OK but I would like to verify the distribution of the stops, there is always something to learn! I exported the optimization values from AmiBroker and produced a 3D-Chart, this is what it looks like.

### StoCCI Profit Stop Analysis



### StoCCI Profit Stop Analysis



From the above charts you can see that letting the profits run above our thresholds of 8.1% (Long) and 9.5% (Short) reduces profits. I find this a very revealing chart; probably worthwhile studying in more detail some other time. After studying the flat chart and then referring back to the 3D chart you will be able to distinguish very distinct features.

### What did we accomplish?

For the period of the initial proposal, 3/1/00 to today, we have:

| System    | Ticker | Net %Profit | Max SysDD | #Trades | Winners | Losers | Exp.   | RAR     | Profit Factor |
|-----------|--------|-------------|-----------|---------|---------|--------|--------|---------|---------------|
| Basic     | INTC   | 143.82%     | -94.00%   | 6       | 5       | 1      | 38.36% | 45.26%  | 18.95         |
| Optimized | INTC   | 2724.03%    | -40.60%   | 118     | 84      | 34     | 34.71% | 236.87% | 2             |

Net profit improved 19 times, approximately in proportion to the number of trades. Draw Down is reduced about 50%. Below is a chart showing both the starting and final

equities for this period. To me one of the main advantages is that this system works on out-of-sample data.

### Before and After



Figure 18 - The starting position (BLK) and result of our efforts (RED)

### Performance Overview

The final equity curve for the 1/1/97 to today period is now:



## The Performance Report

| Overall performance summary    |           |                                      |             |
|--------------------------------|-----------|--------------------------------------|-------------|
| <b>Total net profit:</b>       | 20084.01  | <b>Total commissions paid:</b>       | 0           |
| <b>Return on account:</b>      | 20084.01% | <b>Open position gain/loss</b>       | -821.97     |
| <b>Buy&amp;Hold profit:</b>    | 7.35      | <b>Bars (avg. days) in test:</b>     | 1400 (2032) |
| <b>Buy&amp;Hold % return:</b>  | 7.35%     | <b>System to Buy&amp;Hold index:</b> | 272972.05%  |
| <b>Annual system % return:</b> | 159.44%   | <b>Annual B&amp;H % return:</b>      | 1.28%       |
| <b>System drawdown:</b>        | -20.26    | <b>B&amp;H drawdown:</b>             | -4.55       |
| <b>Max. system drawdown:</b>   | -8744.5   | <b>B&amp;H max. drawdown:</b>        | -361.61     |
| <b>Max. system % drawdown:</b> | -40.60%   | <b>B&amp;H max. % drawdown:</b>      | -78.55%     |
| <b>Max. trade drawdown:</b>    | -6267.08  |                                      |             |
| <b>Max. trade % drawdown:</b>  | -32.88%   |                                      |             |
| <b>Trade drawdown:</b>         | -6172.12  |                                      |             |
| <b>Total number of trades:</b> | 259       | <b>Percent profitable:</b>           | 70.70%      |

|                                    |          |                                   |           |
|------------------------------------|----------|-----------------------------------|-----------|
| Number winning trades:             | 183      | Number losing trades:             | 76        |
| Profit of winners:                 | 41479.61 | Loss of losers:                   | -20573.62 |
| Total # of bars in winners:        | 823      | Total # of bars in losers:        | 527       |
| Commissions paid in winners:       | 0        | Commissions paid in losers:       | 0         |
|                                    |          |                                   |           |
| Largest winning trade:             | 2348.59  | Largest losing trade:             | -5142.6   |
| # of bars in largest winner:       | 2        | # bars in largest loser:          | 15        |
| Commission paid in largest winner: | 0        | Commission paid in largest loser: | 0         |
|                                    |          |                                   |           |
| Average winning trade:             | 226.66   | Average losing trade:             | -270.71   |
| Avg. # of bars in winners:         | 4.5      | Avg. # bars in losers:            | 6.9       |
| Avg. commission paid in winner:    | 0        | Avg. commission paid in loser:    | 0         |
| Max consec. winners:               | 9        | Max consec. losers:               | 3         |
|                                    |          |                                   |           |
| Bars out of the market:            | 250      | Interest earned:                  | 0         |
|                                    |          |                                   |           |
| Exposure:                          | 82.10%   | Risk adjusted ann. return:        | 194.10%   |
| Ratio avg win/avg loss:            | 0.84     | Avg. trade (win & loss):          | 80.72     |
| Profit factor:                     | 2.02     |                                   |           |

## Conclusion

The above showed that with a systematic approach and by taking the time to make some charts you can improve on almost any trading system. The stock and formulas are incidental and you can apply the process on your favorite formulas and stocks.

If you give it a try and the performance of your system improves, please share your experience with me. No need to email formulas, I am only interested to know what difficulties you encountered, what other development techniques you employed, and how you improved on the system.

*Do not trade this system as is; no future profitability is suggested or implied. In fact, I would not trade it since it works only well on INTC.*

The next step is up to you **J** I wish you the best of luck in your trading efforts!  
[Herman](#).